

Une des caractéristiques les plus intéressantes de l'association entre un BASIC et un DOS est la possibilité de créer des fichiers de données gérés par un (ou plusieurs) programmes écrits en BASIC. Pour les possesseurs de Junior Computer, la procédure à suivre est sommairement décrite dans la notice d'Ohio; aussi le programme de gestion bancaire que nous a adressé un lecteur (fort modeste au demeurant) nous est-il apparu comme une excellente occasion de se plonger dans le fonctionnement de ces fichiers indirects.

fichiers indirects avec le Junior Computer

Un programme de gestion bancaire comme exemple d'utilisation des fichiers de données à accès aléatoire du DOS.

L. Germain

Tableau 1. Le programme PRPDA3 ne sert qu'à introduire une fois pour toutes le code (secret) dans le fichier DATA3. Le programme de gestion bancaire proprement dit (GESTIB) fonctionne avec deux fichiers indirects à accès aléatoire DATA2 et DATA3 pour lesquels il faut créer deux zones tampons avant d'introduire les instructions ci-contre.

Pour commencer, ce qu'il faut bien comprendre, c'est la différence fondamentale entre un fichier direct et un fichier indirect. Lorsque le programmeur écrit un programme avec numéros de lignes, lorsqu'il l'édite, le modifie, en fait un listage, il y a un accès direct. Lorsque ce même programme génère lui-même des données, effectue des calculs, compile les résultats et en fait un nouveau fichier (dit "de données") sauvegardé sur disquette sous un autre nom à l'aide des instructions OPEN, CLOSE, GET et PUT, on parle d'accès indirect. En effet, l'utilisateur n'a pas la possibilité d'ajouter, de retirer, voire de lire des données directement dans ce fichier.

Comment préparer les fichiers indirects?

En règle générale, il faut commencer par attribuer des noms aux fichiers. Dans notre exemple, c'est "DATA2" et "DATA3" qui sont les fichiers indirects, "GESTIB" (pour "gestion bancaire") qui est le programme principal et enfin "PRPDA3" (pour "prépare DATA3").

"DATA3" est un fichier indirect à accès aléatoire dont on ne se sert que pour sauvegarder le solde du compte, le numéro du dernier enregistrement (ces deux données sont initialisées automatiquement par "PRPDA3") ainsi qu'un code qui permet d'interdire l'accès au programme à quiconque ne le connaît pas (en remplaçant l'instruction RUN par l'instruction NEW à la fin de la ligne 45 de "GESTIB", le programme s'efface automatiquement si le code est faux!). En début de la ligne 20 de "GESTIB" on peut également rajouter l'instruction POKE 741,10 pour interdire le listage du programme.

"DATA2" contient les opérations bancaires (montants, date et nature) enregistrées par le programme "GESTIB".

Avant d'introduire les programmes "GESTIB" et "PRPDA3" en mémoire, il faut créer des zones tampons à l'aide de l'option 7 de BEXEC*. Pour "GESTIB"

il faut deux zones tampons ("DATA2" et "DATA3") et pour "PRPDA3" il en faut une seule. Au préalable, on aura réservé quatre pistes sur la disquette pour "GESTIB", une pour "DATA3", une pour "PRPDA3" et une dizaine au moins pour "DATA2" (environ un an d'utilisation). Une fois que les tableaux 1 et 2 auront été sauvegardés sur disquette, il faut exécuter "PRPDA3", introduire un code, puis exécuter "GESTIB". Aussitôt que l'on aura introduit le code convenable, on verra apparaître les huit options que nous allons décrire.

Les routines d'utilisation

Au début de chacune des routines spécifiques, le programme va chercher sur la disquette ("DATA3") deux données fondamentales: le solde disponible (variable S) et le numéro du dernier enregistrement effectué (variable C). Cette opération est effectuée à partir de la ligne 500. Le sous-programme à partir de la ligne 525 assure l'opération inverse à la fin de chacune des routines.

Routines d'entrée et de sortie

Le programme demande d'abord quel est le nombre d'opérations de crédit ou de débit à effectuer. Il demande ensuite le montant de la première opération, sa nature (ou catégorie) et sa date.

La date doit toujours être indiquée par 6 chiffres sous la forme JJ/MM/AA, où JJ est le jour, MM le mois et AA l'année. La catégorie est enregistrée sous forme de chaîne de caractères. Il peut donc s'agir de noms ("impôts", par exemple), de numéros (de chèque, par exemple) ou d'abréviations.

Routine de recherches

Il existe la possibilité d'obtenir un listage complet de toutes les opérations de crédit ou de débit, ou un listage par mois, ou encore celui d'une opération précise (identifiée par le montant, la catégorie et la date). Si un ou deux des trois paramètres requis (montant/catégorie/date) sont inconnus, répondre par "X".

Routine de solde

Cette routine affiche le solde disponible.

Routines de totalisation

Comme leur nom l'indique, ces options per-

```
1
5 REM PRPDA3
10 PRINT:PRINT:INPUT"INPUT CODE ";B#:C=0:S=0
20 DISK OPEN,6,"DATA3":DISK GET,4
30 PRINT$6,4;" ";B#:DISK PUT:DISK GET,2
40 PRINT$6,2;" ";C;" ";S:DISK PUT:DISK CLOSE,6
```

```
1 REM GESTIB
5 PRINT:PRINT:PRINT:POKE 2888,0:POKE8722,0
10 PRINTTAB(16)"PROGRAMME DE GESTION BANCAIRE":PRINT:PRINT
20 PRINT:PRINT:INPUT"CODE ";A$:PRINT:PRINT
30 DISK OPEN,6,"DATA3":DISK GET,1
45 INPUT $6,R,B$:DISK CLOSE,6:IFA$<>B$ THEN RUN
70 CLEAR:PRINT:PRINTTAB(3)"1 = ENTREES"
80 PRINTTAB(3)"2 = SORTIES"
90 PRINTTAB(3)"3 = RECHERCHES"
95 PRINTTAB(3)"4 = SOLDE"
96 PRINTTAB(3)"5 = TOTALISATION PAR MOIS"
97 PRINTTAB(3)"6 = TOTALISATION PAR ANNEE"
98 PRINTTAB(3)"7 = TOTALISATION PAR CATEGORIE"
100 PRINTTAB(3)"8 = EXIT"
105 PRINT:INPUT"INDICE CHOISI ";A
110 ON A GOTO 430,200,250,330,340,380,430,4000
120 GOTO70
121 REM
130 GOSUB500:PRINT:PRINT:PRINTTAB(16)"ROUTINE D'ENTREE"
145 PRINT:PRINT:INPUT"NOMBRE DE DONNEES A RENTRER";D:PRINT:PRINT
150 DISK OPEN,6,"DATA2"
155 FOR R=C+1 TO C+D
160 INPUT"MONTANT ";E:S=S+E:PRINT:PRINT
161 INPUT"NUMERO DE L'OPERATION ou CATEGORIE ";A$:PRINT:PRINT
170 INPUT"DATE ";B$:S$="E":PRINT:PRINT:DISK GET,R
180 PRINT$6,R;" ";E;" ";A$;" ";B$;" ";S$
185 DISK PUT:NEXTR:C=R-1:GOSUB525:GOTO70
186 REM
200 GOSUB500:S$="S":PRINT:PRINT:PRINTTAB(16)"ROUTINE DE SORTIE":PRINT:PRINT
205 DISK OPEN,6,"DATA2":INPUT"NOMBRE DE SORTIES ";B:PRINT:PRINT
245 FOR R=C+1 TO C+B
220 INPUT"MONTANT DU RETRAIT ";E:PRINT:PRINT
221 IF(S-E)<0THEN PRINT"IL NE RESTE QUE ";S;" Frs!":PRINT:PRINT:GOTO220
225 S=S-E
226 INPUT"NUMERO DE L'OPERATION ou CATEGORIE ";A$:PRINT:PRINT
227 INPUT"DATE ";B$:PRINT:PRINT
235 DISK GET,R:PRINT$6,R;" ";E;" ";A$;" ";B$;" ";S$
240 DISK PUT:NEXTR:DISK CLOSE,6:GOSUB525:GOTO70
241 REM
250 PRINT:PRINTTAB(16)"ROUTINE DE RECHERCHES":PRINT:PRINT
255 INPUT"RECHERCHE D'UNE ENTREE OU DUNE SORTIE (E/S) ";Z$:PRINT:PRINT
260 INPUT"VOULEZ-VOUS UNE LISTE ";X$:IF X$<>"O" THEN 265:PRINT:PRINT
262 INPUT"VOULEZ-VOUS UNE LISTE PAR MOIS ";H$:PRINT:PRINT:IF H$<>"O" THEN 280
263 GOSUB 500:GOSUB 710:V$=Z$:GOTO345
265 PRINT:PRINT:INPUT"MONTANT ";I$:PRINT:PRINT:IF I$<>"X" THEN 280
270 INPUT"NUMERO DE L'OPERATION OU CAT. ";J$:PRINT:PRINT
271 IF I$<>"X" THEN 280
275 INPUT"DATE ";L$:IF I$="X" THEN PRINT:PRINT"IMPOSSIBILITE !":GOTO70
276 REM
280 GOSUB 500:DISK OPEN,6,"DATA2"
285 FOR R=1 TO C:DISK GET,R
290 INPUT$6,R,E,A$,B$,S$:IF Z$<>S$ THEN 320
295 IF X$="O" THEN GOSUB600:GOTO 320
296 REM
300 IF I$=A$ OR I$=B$ OR E=VAL(I$) THEN GOSUB 600
320 IF W=15 THEN INPUT"CR POUR CONTINUER ";W$:W=0
321 NEXTR:IF Y=0 THEN PRINT:PRINT"DONNEE INCONNUE DANS LE FICHIER"
322 DISK CLOSE,6:Y=0
323 PRINT:PRINT:INPUT"CR POUR CONTINUER ";W$:GOTO70
324 REM
330 GOSUB500:PRINT:PRINT:PRINT"SOLDE DISPONIBLE: ";S;" Frs":GOTO 323
331 REM
340 X$=" MOIS ":GOSUB700
345 PRINT:PRINT:INPUT"SUR QUEL MOIS ";M:DISK OPEN,6,"DATA2"
355 FOR R=1 TO C:DISK GET,R
357 INPUT $6,R,E,A$,B$,S$
360 IF V$<>S$ THEN 365
361 L=VAL(MID$(B$,4,2)):O=VAL(RIGHT$(B$,2))
363 IF H$="O" AND Q=0 AND M=L THEN GOSUB600:IF W=18 THEN 320
364 IF Q=0 AND M=L THEN LET T=T+E
365 NEXTR
370 PRINT:PRINT:PRINT"TOTAL DES ";V$;" DU ";M;"/";Q;" : ";T;" Frs ":T=0:Q=0
375 GOTO 323
376 REM
380 X$="ANNEE ":GOSUB700:DISK OPEN,6,"DATA2"
400 FOR R=1 TO C:DISK GET,R
410 INPUT $6,R,E,A$,B$,S$
415 IF V$<>S$ THEN 425
420 O=VAL(RIGHT$(B$,2))
422 IF Q=0 THEN LET T=T+E
425 NEXTR
427 PRINT:PRINT:PRINT"TOTAL DES ";V$;" DE ";Q;" : ";T;" Frs ":T=0:GOTO 323
428 REM
430 X$=" CATEGORIE ":GOSUB700:PRINT:PRINT:INPUT"CATEGORIE ";I$
435 DISK OPEN,6,"DATA2"
440 FOR R=1 TO C:DISK GET,R:INPUT $6,R,E,A$,B$,S$
442 IF V$<>S$ THEN 450
444 L=VAL(RIGHT$(B$,2)):IF Q=L AND A$=I$ THEN LET T=T+E
450 NEXTR
455 PRINT:PRINT:PRINT"TOTAL DES ";I$;" DE ";Q;" = ";T;" Frs ":T=0:GOTO323
456 REM
500 DISK OPEN,6,"DATA3"
510 DISK GET,2:INPUT $6,R,C,S:DISK CLOSE,6:RETURN
521 REM
525 C=R-1:DISK OPEN,6,"DATA3":DISK GET,2:PRINT$6,2;" ";C;" ";S
540 DISK PUT:DISK CLOSE,6:RETURN
550 REM
600 IF LEN(STR$(E))=9 THEN SP=0:GOTO 620
602 K=LEN(STR$(E)):SP=(9-K)
620 PRINT R;TAB(4)"MONTANT:";SPC(SP);E;"Frs /OPER. OU CAT. ";A$;
621 PRINTTAB(49)"DATE:";B$;" ";S$
630 W=W+1:Y=Y+1:RETURN
700 GOSUB 500:PRINT:PRINT"ROUTINE DE TOTALISATION PAR ";X$
705 PRINT:PRINT:INPUT"SUR ENTREE OU SUR SORTIE (E/S) ";V$
710 PRINT:PRINT:INPUT"ANNEE ";Q:RETURN
1000 END
```

fichiers indirects avec le
Junior Computer
elektor décembre 1983

Tableau 2. Dans le listing du programme GESTIB ci-contre il faut remplacer tous les signes "\$" par le signe "#" dont l'importance est capitale: en effet ce signe, suivi par un numéro, désigne le module (device) mémoire lors du chargement des fichiers indirects et lors d'opérations d'écriture le tampon réservé (par exemple INPUT # 6 ou PRINT # 6). Il suffit d'un seul lecteur de disquettes pour utiliser les fichiers indirects.

mettent la compilation des enregistrements effectués.
Lorsque l'on utilise le programme la première fois, il faut commencer par l'option "1" pour effectuer au moins un crédit, faute de quoi le programme refusera les recherches et les débits.
Il serait trop long de détailler ici toutes les

opérations effectuées par le programme, qu'il a d'ailleurs fallu réduire au strict minimum. Nous sommes persuadés que malgré cela, nos lecteurs intéressés sauront se frayer un chemin à travers les embûches, et ne jureront bientôt plus que par les fichiers indirects.

